
Hands-on Exercise 1 

 
 

Objective: In this exercise, you will add a connection to an Excel Workbook. 

1. Open [W:\Hands-on Exercises\Race, Load, Gender - GRAD ALL.xlsx]. 

 

2. Switch to the [Data] tab. 

 

There are actually two ways to add a connection to an external data source.  Two options are shown 

here.  OPTION 1 works for just Access Databases while OPTION 2 works for all sources.  

OPTION 1 

 

3. Switch to the [DATA] ribbon and select [From Access]. 

 

4. Navigate to [W:\Data\Factbook Data.accdb] and click [Open]. 

 

5. The [Select Table] dialog will appear.  Select [Race, Load, Gender - GRAD ALL] and click [OK]. 

 

 

SKIP TO STEP 15  

Hands-on Exercise 1 

 
 

OPTION 2 

 

6. Switch to the [DATA] ribbon and select [Connections]. 

 

7. Click the [Add…] button to add a new connection. 

 

8. Click the [Browse for More…] button and navigate to [W:\Data\Factbook Data.accdb]. 

9. The [Data Link Properties] dialog may appear.  If so, ensure the following are set correctly and 

click OK. 

Provider Tab 

 OLE DB Provider(s):  Microsoft Office 12.0 Access Database Engine OLE DB Provider 

Connection Tab 

 User Name: Admin 

 Blank password: <Selected> 

Advanced Tab 

 Access permissions: Only Read selected 

Hands-on Exercise 1 

 
 

 

10. The [Select Table] dialog will appear.  Select [Race, Load, Gender - GRAD ALL] and click [OK]. 

 

11. Click the [Close] button. 

 

12. Switch to the [DATA] ribbon and select [Existing Connections]. 

Hands-on Exercise 1 

 
 

13. Section [Factbook Data] and click [Open]. 

 

14. Click [OK] to insert the table on the existing worksheet beginning in cell [$A$1]. 

 

15. The table will be created and appear in the upper left corner of the worksheet (the Rows and 

Columns indicators will have green checkmarks if this step is completed correctly).  

 

Bonus Exercise:   

If you have time, try this exercise on [W:\Hands-on Exercises\Race, Load, Gender - UG ALL.xlsx]. 

Hands-on Exercise 2 

 

Objective: In this exercise, you will insert some formulas referencing a table. 

1. Open [W:\Hands-on Exercises\Race, Load, Gender - GRAD ALL.xlsx]. 

 

2. Switch to the [Data] tab. 

 

3. Click anywhere in the table, switch to the [DESIGN] ribbon under [TABLE TOOLS], and rename the 

table [Factbook_Data]. 

 

4. Switch to the [Process] tab. 

 

5. Correct the formula in cell [A2], replace “#REF!” with “Factbook_Data[Year]”. 

 

6. In cell [B4], enter “=COUNTIFS(Factbook_Data[Gender],$A4,Factbook_Data[Race],B$3)”.  Copy and 

paste cell [B4] (DO NOT DRAG AND FILL) to cells [C4] – [J4].  Enter “=SUM(B4:J4)” into cell [K4]. 

 

Check answers in cells [B28] – [K28]. 

 

7. Copy cell [B4] to [B6].  Notice the result is now “0”.  Edit the formula and make sure it is as follows: 

 

“=COUNTIFS(Factbook_Data[Gender],$A4,Factbook_Data[Race],B$3,Factbook_Data[Load],$A6)” 

 

Copy and paste cell [B6] to cells [C6] – [J6] and copy cell [K4] to [K6]. 

 

8. Copy and paste cell [B6] to cell [B8] and correct the formula so the Gender criteria is coming from 

[$A4] instead of [$A6].  

 

Copy and paste cell [B8] to cells [C8] – [J8] and copy cell [K6] to [K8]. 

 

9. Complete the Female section in the same way and use SUM to calculate the grand totals. 

 

Bonus exercise: Repeat this process for [W:\Hands-on Exercises\Race, Load, Gender - UG ALL.xlsx]. 

Hands-on Exercise 3 

Objective: In this exercise, you will insert some formulas into a worksheet. 

1. Open [W:\Hands-on Exercises\KPIs.xlsm]. 

 

2. Switch to the [Retention Rate] tab. 

 

3. In cell [V3], enter “=MATCH("Actual",$A:$A)”.   

 

4. In cell [U7], enter “=AVERAGE(OFFSET($A$1,$V$3-1,$W$3-3,1,3))”. 

 

5. Copy cell [U7] into cell [U8] and change the formula to use 10 columns of data instead of 3  

(Hint: two of the 3’s in the formula above need changed to 10’s – which 3’s will you choose?). 

 

Bonus exercise: Complete cells [U5] and [U6]. 

Hands-on Exercise 4 

Objective: In this exercise, you will insert some formulas into a worksheet and use Drag and Fill. 

1. Open [W:\Hands-on Exercises\Credit Hours - GRAD ALL.xlsx]. 

 

2. Switch to the [Process] tab. 

 

3. Enter “=VLOOKUP($A3,Data[#All],COLUMN(B2))” into cell [B3].   

 

4. Drag and Fill from cell [B3] to cells [C3] and [D3]. 

 

5. Drag and Fill from cells [B3:D3] to cells [B4] through [D7]. 

Hands-on Exercise 5 

Open [W:\Hands-on Exercises\ Program Prioritization.xlsm]. 

We will complete the highlighted row of cells, summing degrees within the same CIP, level, and 

institutuon.  Use a SUMIFS formula. 

 

 

We are going to start by finding all majors for a given institution.   Here is your starting formula:   

 

=SUMIFS(Enrollment_Conferral[2007_Majors], Enrollment_Conferral[Institution],'Enrollment 

and Degree - Process'!A16) 

 

 

In this example, “Enrollment Conferral” is the table name.  It’s located on the “Enrollment and Degree - 

GA Data” tab. 

 

TO BUILD THIS FORMULA: 

Start by type =SUMIFS( 

 

Table name Field name Table name Field name 

Criteria Value - Institution 

Hands-on Exercise 5 

 

This handy box comes up to prompt the user about the components and syntax of the formula. 

 

 

Notice that the first portion of the formula is the sum_range.  That is the data that will actually be 

summed.  

Navigate to the “Enrollment and Degree - GA Data” tab.  Click directly above the header of the column 

to be summed.  In this case, that is 2007 Majors. 

 

 

That click will complete the first part of the formula.   

Next, the formula calls for the criteria range 1, and then criteria 1.  In this case, the criteria range is the 

column with data about institution.   

 

In your formula, insert a comma, and then click the top of the institution column.  Your formula should 

now look like this: 

 

 

Again, put a comma in your formula to move to the next part – the criteria 1.  In this case, the criteria 1 

is “NCCU”.  You formula would be correct if you typed the abbreviation directly: 

=SUMIFS(Enrollment_Conferral[2007_Majors],Enrollment_Conferral[Institution],"NCCU") 

Hands-on Exercise 5 

But the better approach is to fill the criteria with the cell reference, so it can be reused for UNC-P, UNC-

W, etc. 

If you already typed “NCCU”, just delete that part of your formula, and click the cell with the NCCU 

abbreviation. 

 

In this case, the new reference A16 isn’t anchored at all.  That means that if you drag that formula to the 

next cell, it will change the reference to B16.  Take a moment to try that if you like. 

We can add an anchor to that reference, so that it always looks in column A.  To do this, use the F4 key 

to cycle through the anchor options (i.e., anchor both row and column, only row, only column, neither) 

until you have made the reference $A16 (anchored column, moveable row). 

Your formula at this point should look like this: 
=SUMIFS(Enrollment_Conferral[2007_Majors],Enrollment_Conferral[Institution],'Enrollment and Degree 
- Process'!A16) 
 
Now, we still need to add the criteria for level.  To do this, click into the parenthesis at the end of the 
formula, and add the criteria range for level, and then the criteria for level. 
 
Your formula at this point should look like this: 
 
=SUMIFS(Enrollment_Conferral[2007_Majors],Enrollment_Conferral[Institution],'Enrollment and Degree 
- Process'!$A16,Enrollment_Conferral[Degree_Level],'Enrollment and Degree - Process'!$B$3) 
 
Notice above that $B$3 is anchored on both the row and column.  This is because the level value is 
always in the exact same cell. 

Hands-on Exercise 5 

Now add the final criteria – CIP code.  Same as before, add the criteria range (i.e., column of CIP codes, 
followed by the criteria (i.e., CIP code 540101). 
 
Now your formula should look like this: 
=SUMIFS(Enrollment_Conferral[2007_Majors],Enrollment_Conferral[Institution],'Enrollment and Degree 
- Process'!$A16,Enrollment_Conferral[Degree_Level],'Enrollment and Degree - 
Process'!$B$3,Enrollment_Conferral[CIP_Code],'Enrollment and Degree - Process'!$B$4) 
 
In plain English that formula reads: 
=SUMIFS(2007 majors, Institution, “NCCU”, Level, “Bachelor’s”, CIP Code, 540101) 
 
 
Now, fill in the rest of that row.  The icons to the right will indicate if you have achieved the correct 
answer. 
Note – In this case, degree for each year are in a separate column, so adjust your formulas accordingly. 

Hands-on Exercise 6 

Open [W:\Hands-on Exercises\ Program Prioritization.xlsm]. 

Now we will move onto an example using Index-Match, within an IF statement. 

At this point, your sheet should look like the screenshot below, and we are now going to work on the 

highlighted cells, which provide percentage of total enrollment within the level.    (Rationale – our 

leadership wanted to look at numbers of majors within a CIP and level at other institutions, but wanted 

those numbers to be within the context of institution size) 

  

We will start with the INDEX-MATCH part of the formula. 

Here is the guidance Excel provides when we type =INDEX( 

 

This is one of the least clear formula descriptions that Excel provides, mainly because the INDEX formula 

is versatile and can be used in several ways.  Don’t be discouraged, we will walk through an example. 

In this case, we will start by pulling the total institutional enrollment for the year into the cell.  In plain 
English, here is the first step: 
=INDEX(dimensions of the reference table) 
 
Here we are going to reference a table where I have pre-totaled the enrollment by institutions by year.  

That table name is “Enrollment_Conferrals_Total” and it’s also on the “Enrollment and Degree - GA 

Data” tab. 

Hands-on Exercise 6 

==INDEX('Enrollment and Degree - GA Data'!R:X) 

Now we will add the MATCH formula, which will find a specific value in the INDEX table. 

Here is Excel’s description of the MATCH function: 

 

In this case, the lookup value is the institution.  Below you will see that we can indicate the cell 

reference that contains the NCCU abbreviation, and anchor on the column (as the institution 

abbreviation is always in the column A): 

=INDEX('Enrollment and Degree - GA Data'!R:X, MATCH('Enrollment and Degree - Process'!$A16) 

The Lookup array part of the MATCH formula is looking for the column of the INDEX table that contains 

the matching value (in this case the institution abbreviation). 

=INDEX('Enrollment and Degree - GA Data'!R:X, MATCH('Enrollment and Degree - 

Process'!$A16),'Enrollment and Degree - GA Data'!R:R) 

The final part of the formula (match type) = 0 for an exact match, so at the end of the MATCH formula 

above, we will put a comma, followed by a zero, to indicate that we are looking to match the institution 

abbreviation exactly. 

=INDEX('Enrollment and Degree - GA Data'!$R:$X, MATCH('Enrollment and Degree - 

Process'!$A16,'Enrollment and Degree - GA Data'!$R:$R, 0) 

Finally, once you close the parenthesis on the MATCH formula, you will see that one additional 

argument is needed for the INDEX, namely the column number.  In this case the column is 2 (that 

column which holds total enrollment for 2007). 

At this point, you should have a crazy number – 838300% 

Next we need to divide the program enrollment into the total enrollment.  This is easy – at the very start 

of the formula, just drop in the cell reference with the problem enrollment about, and divide by the 

calculation you just made: 

=B16/INDEX('Enrollment and Degree - GA Data'!$R:$X, MATCH('Enrollment and Degree - 

Process'!$A16,'Enrollment and Degree - GA Data'!$R:$R, 0),2) 

One more thing to think about, if your program enrollment for this combination of institution, CIP, and 

level is zero (a particular institution didn’t offer a CIP, at a certain level, in a certain year), your formula 

will return an error.  You can avoid this error printing on your report by checking for a zero, and 

suppressing the error in cases of zero.  To do this, at the beginning of your formula type  

Hands-on Exercise 6 

=IF( 

Now evaluate the value of the enrollment cell: 

 

If the enrollment figure is greater than 0, proceed with the calculation, if not, return a null value 

Hands-on Exercise 7 

Open [W:\Hands-on Exercises\ Program Prioritization.xlsm]. 

Now we will move onto an example using an array formula that was built to manage a “MEDIANIFS” 

calculation.  This example is on the same tab as the last example - Enrollment and Degree – Process 

 

 

We knew the functionality we were looking for, and found an example on the web that we followed.  

This is a complex example, and there really isn’t a way to build the formula incrementally, so we will 

start with this: 

=MEDIAN(IF(Enrollment_Conferral[Degree_Level]='Enrollment and Degree - Process'!$R7, 

Enrollment_Conferral[2007_Majors])) 

NOTE – Instead of pasting the formula, then clicking ENTER, you will click CTRL+SHIFT+ENTER 

simultaneously.  Sometimes these formulas are referred to as CSE for this reason.  That action should 

add curly braces around the formula, so that it looks like this.  (PS – You cannot type the curly braces 

directly, and get it to work; you need to let Excel change the formula on the CSE action.) 

{=MEDIAN(IF(Enrollment_Conferral[Degree_Level]='Enrollment and Degree - Process'!$R7, 

Enrollment_Conferral[2007_Majors]))} 

At this point, the result of the calculation should be 48.  This is median size of all bachelors’ programs in 

the data set.   

 

In this case, we want to get only the median size of WCU programs, so we will add the WCU criteria, 

resulting in this formula (the added parts are highlighted below): 

{=MEDIAN(IF((Enrollment_Conferral[Degree_Level]='Enrollment and Degree - 

Process'!$R7)*(Enrollment_Conferral[Institution]="WCU"), Enrollment_Conferral[2007_Majors]))} 

This step checks for both values to be true (i.e., level = bachelor’s AND institution = WCU), so it is a trick 

for adding multiple criteria into the formula. 

Hands-on Exercise 7 

Finally, as this will occasionally return an error, we can add an IFERROR statement to handle cases where 

a particular institution did not have any enrollment at a particular level in a particular year. 

The final formula is  

{=IFERROR(MEDIAN(IF((Enrollment_Conferral[Degree_Level]='Enrollment and Degree - 

Process'!$R7)*(Enrollment_Conferral[Institution]="WCU"), 

Enrollment_Conferral[2007_Majors])),"N/A")} 

 

This is quite complex, and took us a little while to figure out.  The main purpose of this exercise is to 

introduce the concept, so that you know this expanded functionality exists within Excel.  Please don’t 

get discouraged if this doesn’t make sense to you yet! 

